

DATA GENERAL
CORPORATION

Southboro,
Massachusetts 01772
(617) 485-9100

PROGRAM

Bootstrap Loader

TAPES

Not applicable

ABSTRACT

The Bootstrap Loader is a short routine used to load the Binary Loader into memory.

1. REQUIREMENTS

1.1 Memory

1K or larger alterable memory

1.2 Equipment

Teletype ASR or paper tape reader

1.3 External Subroutines

None

1.4 Other

None

2. OPERATING PROCEDURE

2.1 Calling Sequence

The Bootstrap Loader must be keyed into memory from the NOVA control console. The listings at the end of this write-up give the octal encoding for the instructions that make up the routine. The listings are originated at 07757 for illustrative purposes. The user should key the Bootstrap into memory starting at location X757 where "X" represents the two most significant octal digits of the highest memory address for his system. For example, X = 07 for a 4K system and 17 for an 8K system. The version which is keyed in depends upon the input device to be used, i.e. teletype or paper tape reader.


The Bootstrap is started by entering X770 in the data switches, pressing RESET, and pressing START.

2.2 Input Format


The input for the Bootstrap Loader is a specially formatted tape of the Binary Loader, 091-000004. This tape should be mounted in the input device before starting the Bootstrap.

The format of the input tape is:

Channel 87654321



Channels 8, 7, 6, are never punched. Channel 5 is punched at the beginning of a group of four frames that are assembled to form a 16-bit word. The word assembles as shown below.



The letters correspond to information punched in channels 4 through 1.

2.3 Output Format

The Binary Loader will be loaded into memory relative to the Bootstrap Loader. At completion, the last 131 (octal) words of memory will contain both the Binary and the Bootstrap Loaders.

2.4 Error Returns

No error checking is done by the Bootstrap Loader. This requires the input device to be functioning properly and the input tape to be correctly punched in order to insure proper loading of the Binary Loader.

2.5 State of Active Registers upon Exit

Not applicable

2.6 Cautions to User

The Binary and Bootstrap Loaders require the last 131 (octal) words of memory. The user should write routines which do not use these words as storage. This will allow the loaders to remain in memory and not to be destroyed during a user run.

3. DISCUSSION

3.1 Algorithms

The Bootstrap reads the special format input tape and assembles 16-bit words as described in 2.2. The first two words, decoded as instructions, are STA 1, .+1 and JMP .-4. These instructions are stored in the last two locations of the Bootstrap as indicated in the listing. The third, fifth, ... words are STA instructions using AC1. The fourth, sixth, ... words are data (specifically the Binary Loader instruction words). The Bootstrap executes each odd-numbered word to store the succeeding data word in the location specified by the STA instruction. The final odd-numbered word is a HALT which terminates the routine with X775 displayed in the ADDRESS lamps.

3.2 Limitations and Accuracy

Not applicable

3.3 Size and Timing

The Bootstrap requires 15 (octal) instruction words and 2 temporary locations.

Its speed is limited by the input device.

3.4 References

See write-up 093-000003 for a description of the Binary Loader.

See Section 2.7 of "How to Use the NOVA" for the procedure of keying information from the console into memory.

3.5 Flow Diagrams

Not applicable

4. EXAMPLES AND APPLICATIONS

None

5. PROGRAM LISTINGS

Two listings of the Bootstrap Loader are given. The first version should be used if input is from the teletype reader. The second version should be used if input is from the paper tape reader.

;BOOTSTRAP LOADER (TELETYPE VERSION)
 ;THIS LOADER IS PLACED IN MEMORY MANUALLY
 ;USING THE COMPUTER CONSOLE SWITCHES
 ;IT IS USED TO LOAD THE BLOCK BINARY
 ;LOADER INTO MEMORY.

;THIS LOADER IS NOT DISTURBED BY ANY STANDARD
 ;PROGRAMS AND SHOULD REMAIN IN CORE UNLESS
 ;DISTURBED BY AN UNDEBUGGED USER PROGRAM.

;IT MAY BE PLACED ANYWHERE IN CORE AND THE
 ;BLOCK BINARY LOADER WILL BE LOADED RELATIVE
 ;TO IT.

;NORMALLY, IT RESIDES IN THE UPPER LOCATIONS
 ;OF THE MEMORY SYSTEM.

;IT IS SHOWN HERE ASSEMBLED WITH AN ORIGIN OF 7757
 ;AND HENCE IT IS SHOWN FOR A 4K SYSTEM.

;TO OPERATE:

; 1. PLACE BINARY BLOCK LOADER TAPE IN READER
 ; AND SET IT TO START
 ; 2. PRESS RESET
 ; 3. SET SWITCHES TO 07770 (FOR 4K SYSTEM)
 ; 4. PRESS START
 ; 5. BINARY BLOCK LOADER WILL BE LOADED
 ; AND COMPUTER WILL HALT
 ; WITH ADDRESS=07775 (FOR 4K SYSTEM).

007757 .LOC 7757

07757	126440	GET:	SUBO 1,1	;CLEAR AC1,CRY
07760	063610		SKPDN TTI	
07761	000777		JMP --1	;WAIT FOR DONE FLAG
07762	060510		DIAS 0,TTI	;READ INTO AC0 AND RESTART READ
07763	127100		ADDL 1,1	;SHIFT AC1 LEFT
07764	127100		ADDL 1,1	;4 PLACES
07765	107003		ADD 0,1,SNC	;ADD IN THE NEW WORD
07766	000772		JMP GET+1	;FULL WORD NOT ASSEMBLED YET
07767	001400		JMP 0,3	;OK, EXIT

;BOOTSTRAP LOADER STARTS HERE

07770	060110	BSTRP:	NIOS TTI	;START THE READER
07771	004766		JSR GET	;GET A WORD
07772	044402		STA 1,++2	;STORE IT TO EXECUTE IT
07773	004764		JSR GET	;GET ANOTHER WORD
				;THIS WILL CONTAIN A STA INSTR
				;THIS WILL CONTAIN JMP --4

.END

;BOOTSTRAP LOADER (HIGH SPEED READER VERSION)
 ;THIS LOADER IS PLACED IN MEMORY MANUALLY
 ;USING THE COMPUTER CONSOLE SWITCHES
 ;IT IS USED TO LOAD THE BLOCK BINARY
 ;LOADER INTO MEMORY.

;THIS LOADER IS NOT DISTURBED BY ANY STANDARD
 ;PROGRAMS AND SHOULD REMAIN IN CORE UNLESS
 ;DISTURBED BY AN UNDERBUGGED USER PROGRAM.

;IT MAY BE PLACED ANYWHERE IN CORE AND THE
 ;BLOCK BINARY LOADER WILL BE LOADED RELATIVE
 ;TO IT.

;NORMALLY, IT RESIDES IN THE UPPER LOCATIONS
 ;OF THE MEMORY SYSTEM.

;IT IS SHOWN HERE ASSEMBLED WITH AN ORIGIN OF 7757
 ;AND HENCE IT IS SHOWN FOR A 4K SYSTEM.

;TO OPERATE:

1. PLACE BINARY BLOCK LOADER TAPE IN READER
AND SET IT TO START
2. PRESS RESET
3. SET SWITCHES TO 07770 (FOR 4K SYSTEM)
4. PRESS START
5. BINARY BLOCK LOADER WILL BE LOADED
AND COMPUTER WILL HALT
WITH ADDRESS=07775 (FOR 4K SYSTEM).

007757 .LOC 7757

07757	126440	GET:	SUBO 1,1	;CLEAR AC1,CRY
07760	063612		SKPDN PTR	
07761	000777		JMP --1	;WAIT FOR DONE FLAG
07762	060512		DIAS 0,PTR	;READ INTO AC0 AND RESTART READ
07763	127100		ADDL 1,1	;SHIFT AC1 LEFT
07764	127100		ADDL 1,1	;4 PLACES
07765	107003		ADD 0,1,SNC	;ADD IN THE NEW WORD
07766	000772		JMP GET+1	;FULL WORD NOT ASSEMBLED YET
07767	001400		JMP 0,3	;OK, EXIT

;BOOTSTRAP LOADER STARTS HERE

07770	060112	BSTRP:	NIOS PTR	;START THE READER
07771	004766		JSR GET	;GET A WORD
07772	044402		STA 1,++2	;STORE IT TO EXECUTE IT
07773	004764		JSR GET	;GET ANOTHER WORD
				;THIS WILL CONTAIN A STA INSTR
				;THIS WILL CONTAIN JMP --4

.END